
  
//  ****************************************************************************************

void Taste()
  {
    // Right, Up, Down, Left, Select, 5V -->  Analogwert 0, 99, 256, 409, 639, 1023 (Digits)
    TasteAnalogIn = analogRead(0);            // Abfrage Tasten auf analogen Eingang
    //  Variante 1, wenn man gut im Kopfrechnen ist:  
      if (TasteAnalogIn >= 831) TasteNummerNeu = 0;    // 639 + ( 1023 - 639 ) / 2 = 831
      if (TasteAnalogIn <  831) TasteNummerNeu = 1;    // 639 + ( 1023 - 639 ) / 2 = 831
      if (TasteAnalogIn <  524) TasteNummerNeu = 2;    // 409 + (  639 - 409 ) / 2 = 524
      if (TasteAnalogIn <  333) TasteNummerNeu = 3;    // 256 + (  409 - 256 ) / 2 = 332,5
      if (TasteAnalogIn <  178) TasteNummerNeu = 4;    //  99 + (  256 -  99 ) / 2 = 177,5
      if (TasteAnalogIn <   50) TasteNummerNeu = 5;    //   0 + (   99 -   0 ) / 2 =  49,5

//    Variante 2, wenn der Compiler rechnen soll:    
//      if (TasteAnalogIn >= 639 + ( 1023 - 639 ) / 2) TasteNummerNeu = 0;    //  ------
//      if (TasteAnalogIn <  639 + ( 1023 - 639 ) / 2) TasteNummerNeu = 1;    //  Select
//      if (TasteAnalogIn <  409 + (  639 - 409 ) / 2) TasteNummerNeu = 2;    //  Left
//      if (TasteAnalogIn <  256 + (  409 - 256 ) / 2) TasteNummerNeu = 3;    //  Down
//      if (TasteAnalogIn <   99 + (  256 -  99 ) / 2) TasteNummerNeu = 4;    //  Up
//      if (TasteAnalogIn <    0 + (   99 -   0 ) / 2) TasteNummerNeu = 5;    //  Right

    // Taste wurde betätigt
      if (TasteNummerAlt != TasteNummerNeu && TasteNummerNeu != 0)
        {
          // Uhr synchronisieren auf 0:00 und entsprechendes Call auswählen
            if (TasteNummerNeu == 1)
              {
                setTime(0, 0, 0, 20, 07, 2017);
                cntNeuCall = 0;
              }

          // Bandwechsel -/+            
            if (TasteNummerNeu == 3 || TasteNummerNeu == 4)
              {
                  if (TasteNummerNeu == 3)
                    {
                      cntNeuBand--;
                        if (cntNeuBand < 0) {cntNeuBand=4;}
                    }
                   else
                    {
                      cntNeuBand++;
                        if (cntNeuBand > 4) {cntNeuBand=0;}
                    }
                delay(250);                   // Wiederholrate Bandwechsel
              }
        }
  }
        
//  ****************************************************************************************

void ZeitLesen()                              // Softwaretimer Minute und Sekunde lesen
  {
    // Softwaretimer Minute und Sekunde lesen
    istSekunde = second();
    istMinute = minute() % 3;
  }
  
//  ****************************************************************************************

void ZeitAusgabe()                            // Anzeige Zeit aktualisieren
  {
    // Zeit ausgeben
      if (altMinute != istMinute)
        {



          // Neuen Zeitwert Minute schreiben
          lcd.setCursor(12, 1);               // 13. Position in 2. Zeile
          lcd.print(istMinute);
          lcd.print(":");
        }

      if (altSekunde != istSekunde)
        {
          // Neuen Zeitwert Sekunde schreiben
          lcd.setCursor(14, 1);               // 15. Position in 2. Zeile
          if (istSekunde <=9) lcd.print(F("0"));
          lcd.print(istSekunde);
        }
  }
  
//  ****************************************************************************************

void BeaconAktiv()                            // Aktive Beacon ermitteln
  {
    //  Im Raster von 10 Sekunden aktive Station (Calsign) ausgeben 
      if (istSekunde % 10 == 0 && istSekunde != altSekunde)
        { 
          cntTimeCall = (istSekunde / 10 + istMinute % 3 * 6);
          // Beacon zu Zeit berechnen, Beacon auf14,100MHz ist führend
          cntNeuCall = cntTimeCall % 18;
        }
  }

//  ****************************************************************************************

void BandWechsel()                            // Frequenz aktualisieren
  {
      if (cntAltBand != cntNeuBand)
        {
          // neue Frequent schreiben
          lcd.setCursor(7, 0);                // 8. Position in 1. Zeile
          lcd.print(band[cntNeuBand]);
          lcd.print("  ");
        }
  }

//  ****************************************************************************************

void BeaconWechsel()                          // Beacon Call und Ort aktualisieren
  {
      if (cntAltCall != cntNeuCall || cntAltBand != cntNeuBand)
        {
          // neues Rufzeichen ausgeben
          lcd.setCursor(0, 0);                // 1. Position in 1. Zeile
          lcd.print(call[(18+cntNeuCall-cntNeuBand)%18]);
          lcd.print(" ");

          // neuen Ort ausgeben
          lcd.setCursor(0, 1);                // 1. Position in 2. Zeile
          lcd.print(country[(18+cntNeuCall-cntNeuBand)%18]);
        }
  }

//  ****************************************************************************************

void Aktualisierung()                         // Dynamische Werte aktualisieren
  {
    // Dynamische Werte aktualisieren
    TasteNummerAlt      = TasteNummerNeu;     // Tastennummer
    cntAltBand          = cntNeuBand;         // Bandwechsel
    cntAltCall          = cntNeuCall;         // Callwechsel
    altSekunde          = istSekunde;         // Zeit Sekunde



    altMinute           = istMinute;          // Zeit Minute
  }

//  ****************************************************************************************


